
Implementing the
Forth Inner

Interpreter in
High Level Forth

Introduction

Threaded Code
Number literals

Printing string literals

Control structures

Limitations

Implementation
Inner Interpreter

Return stack operations

Inline number literals

Printing inline string literals

Control structures

Compiler to threaded code

Interpreter in threaded code

Conclusion

Demo

U. Hoffmann
Kap. 1 / 22

EuroForth 2016, Reichenau/Konstanz

U. Hoffmann <uh@fh-wedel.de>
FH Wedel

Implementing the Forth Inner Interpreter in High
Level Forth



Implementing the
Forth Inner

Interpreter in
High Level Forth

Introduction

Threaded Code
Number literals

Printing string literals

Control structures

Limitations

Implementation
Inner Interpreter

Return stack operations

Inline number literals

Printing inline string literals

Control structures

Compiler to threaded code

Interpreter in threaded code

Conclusion

Demo

U. Hoffmann
Kap. 2 / 22

Overview

Introduction

Threaded Code
Number literals
Printing string literals
Control structures
Limitations

Implementation
Inner Interpreter
Return stack operations
Inline number literals
Printing inline string literals
Control structures
Compiler to threaded code
Interpreter in threaded code

Conclusion

Demo



Implementing the
Forth Inner

Interpreter in
High Level Forth

Introduction

Threaded Code
Number literals

Printing string literals

Control structures

Limitations

Implementation
Inner Interpreter

Return stack operations

Inline number literals

Printing inline string literals

Control structures

Compiler to threaded code

Interpreter in threaded code

Conclusion

Demo

U. Hoffmann
Kap. 3 / 22

Introduction

I development of hard real time system
I desire to have an interactive command shell while still meeting

the real time requirements
I idea: have a controllable Forth interpreter (inner & outer)

stepwise execution

Threaded Code

I words of underlying system become primitives
I only threaded code no structure of header/dictionary

~: sq ( x - x ) dup * ~;

body
of sq
|-----------+---------+-------------|
| xt of dup | xt of * | xt of ~exit |
|-----------+---------+-------------|

^
|
|

|----|
| IP | Interpreter pointer
|----|

Number literals

~: test ( - u ) 3 4 + ~;

body
of test
|------------+---+------------+---+---------+-------------|
| xt of ~lit | 3 | xt of ~lit | 4 | xt of + | xt of ~EXIT |
|------------+---+------------+---+---------+-------------|

Printing string literals

~: test3 ( - ) ~." it works" ~;

body
of test3 aligned

|------------+---+-----+-----+-----+-----+-----+---+-------------|
| xt of (~." | 8 | ’i’ | ’t’ | ’ ’ | ... | ’s’ | | xt of ~exit |
|------------+---+-----+-----+-----+-----+-----+---+-------------|

Control structures

: looptest ( -- )
0 10
~BEGIN
1- dup

~WHILE
swap over + swap

~REPEAT drop ~;

|------------+---+------------+----|
A | xt of ~lit | 0 | xt of ~lit | 10 |

|------------+---+------------+----|

|----------+-----------+----------------+--------------|
B | xt of 1- | xt of dup | xt of ~?branch | address of D |

|----------+-----------+----------------+--------------|

|---------+------------+---------+------------+---------------+--------------|
C | xt swap | xt of over | xt of + | xt of swap | xt of ~branch | address of B |

|---------+------------+---------+------------+---------------+--------------|

|---------+-------------|
D | xt drop | xt of ~exit |

|---------+-------------|

Limitations

I (user) variables or constants.
I DO LOOPs.
I Neither defining words nor DOES>
I a primitive for pushing address and length of string literals on the

stack (like S").

Implementation

I Inner Interpreter
I Return Stack
I Compiler to Threaded Code
I Outer Interpreter in Threaded Code

Inner Interpreter

Variable IP 0 IP !

\ Perform a single interpretation step
: step ( i*x -- j*x )

IP @ dup cell+ IP !
@ catch
?dup IF cr ." Error " . reset THEN ;

\ Loop steps
: run ( i*x -- j*x )

BEGIN IP @ WHILE step REPEAT ;

Interpreter State

I Interpreter Pointer IP:

Variable IP 0 IP !

I ~R0 with corresponding return stack pointer RP.

Create ~R0 20 cells allot
Variable RP ~R0 RP !

I a data stack shared with the underlying system
I memory (code and data) also shared with the underlying system.

Return stack operations

Create ~R0 20 cells allot
Variable RP ~R0 RP !

: ~>r ( x -- )
\ push a cell to the return stack
RP @ ! 1 cells RP +! ;

: ~r> ( -- x )
\ pop a cell from the return stack
-1 cells RP +! RP @ @ ;

Inline number literals

|----|
| IP |--------------+......
|----| | .

| .
V V

|-----+------------+-----+-----|
| ... | xt of ~lit | val | ... |
|-----+------------+-----+-----|

: ~lit ( -- n )
\ extract inline number literal
IP @ @ 1 cells IP +! ;

Printing inline string literals

|----|
| IP |--------------+...........................
|----| | .

| .
V V

|-----+------------+-----+----+----+-----+----+-----|
| ... | xt of (~." | len | c1 | c2 | ... | cn | ... |
|-----+------------+-----+----+----+-----+----+-----|

: ~." ( <ccc>" -- )
\ Compile inline string to be
\ printed later when executed.
\ Like ." but for threaded code
[’] (~." ,
[char] " word count
here over 1+ chars allot place align

; immediate

Control structures

: ~branch ( -- )
\ absolute unconditional jump
IP @ @ IP ! ;

: ~?branch ( f -- )
\ absolute conditional jump
IF 1 cells IP +!
ELSE ~branch THEN ;

: ~IF ( -- x ) [’] ~?branch , here 0 , ; immediate
: ~AHEAD ( -- x ) [’] ~branch here 0 , ; immediate
: ~ELSE ( x -- x’ ) [’] ~branch , here 0 , swap here swap ! ; immediate
: ~THEN ( x -- ) here swap ! ; immediate

: ~BEGIN ( -- x ) here ; immediate
: ~WHILE ( x1 -- x2 x1 ) [’] ~?branch , here 0 , swap ; immediate
: ~AGAIN ( x -- ) [’] ~branch , , ; immediate
: ~UNTIL ( x -- ) [’] ~?branch , , ; immediate
: ~REPEAT ( x2 x1 -- ) postpone ~AGAIN postpone ~THEN ; immediate

Compiler to threaded code

variable ~state ~state off \ threaded code outer interpreter state

-13 Constant #notfound

: ~] ( -- ) ~state on
BEGIN ( )

BEGIN ( )
bl word dup c@ \ scan next token

WHILE ( c-addr ) \ another token found
find ?dup \ look up in dictionary
IF -1 = IF , ELSE execute THEN ~state @ 0= IF EXIT THEN \ found
ELSE 0 0 rot count

over c@ [CHAR] - = dup >r IF 1- swap char+ swap THEN \ word not found
>number IF #notfound throw THEN
drop drop r> IF negate THEN
[’] ~lit , , \ compile threaded code literal

THEN
REPEAT ( c-addr ) \ no more tokens in input stream
DROP
SOURCE-ID 0= IF CR ." ] " THEN
REFILL 0= \ read more from input stream

UNTIL ; \ input stream exhausted

Compiler to threaded code

: ~[ ( -- ) ~state off ; immediate \ stop threaded code compiler

: ~: ( <name> -- )
\ push IP to return stack and set IP to start of threaded code.
Create ~] Does> IP @ ~>r IP ! ;

: ~EXIT ( -- )
\ Pop IP from return stack
~r> IP ! ;

: ~; ( -- )
\ Compile end of definition and leave threaded code outer compiler
[’] ~EXIT , ~state off ; immediate

Compiler to threaded code

~: name .... ~;

name run

~: sq ( x - x ) dup * ~;

body
of sq
|-----------+---------+-------------|
| xt of dup | xt of * | xt of ~exit |
|-----------+---------+-------------|

^
|
|

|----|
| IP | Interpreter pointer
|----|

Interpreter in threaded code

~: ~interpret ( -- )
~BEGIN ( )
bl word dup c@ \ scan next token

~WHILE ( c-addr ) \ another token found
find \ lookup in dictionary
dup 1 = ~IF drop execute ~ELSE \ immediate
dup -1 = ~IF drop state @ ~IF compile, ~ELSE execute ~THEN ~ELSE
\ word not found, number?
drop 0 0 rot count over c@ 45 = dup ~>r ~IF 1- swap char+ swap ~THEN

>number ~IF #notfound throw ~THEN drop drop \ maybe number
~r> ~IF negate ~THEN
state @ ~IF postpone LITERAL ~THEN \ compile literal

~THEN ~THEN
~REPEAT ( c-addr )

drop ~;

~: ~quit ( -- ) clear-stack ~R0 RP ! ~state off interpret-mode
~BEGIN cr state @ ~IF ~." ] " ~THEN ~query ~interpret ~." ~ok" ~AGAIN ~;

~quit run

Conclusion

I threaded code structure for Forth colon definitions
I inner interpreter, step, in high level Forth.
I a compiler to generate this threaded code
I an interactive outer interpreter in threaded code controllable by

step.
I interactive command interpreter in hard real time system

Demonstration

Demo

